

David Fanning의 ncdf_browser

IDL은 netCDF 를 읽고 쓰기 위한 자체적인 라이브러리를 가지고 있습니다. 이는 netCDF 포맷을 개발한 Unidata 의 C 언어용 netCDF 라이브러리를 그대로 IDL 에 심은 것입니다. netCDF나 HDF 포맷은 자체적으로 파일을 설명하는 다양한 장치들이 내포되어 있지만, 처음부터 직접 netCDF 라이브러리를 이용하여 이런 정보를 조사하는 것은 좀 번거로울 수 있습니다.

IDL 사용자에게 항상 많은 도움을 주고 계신 Fanning 박사의 NCDF_BROWSER 라는 프로그램을 한번 사용해 보시길 권장합니다. 파일을 조사하는 것은 물론, 필요한 변수를 바로 IDL 로 읽어 들일 수 있습니다.

NCDF_BROWSER는 D. Fanning 박사의 유명한 coyote 라이브러리에 함께 들어 있습니다. 만일 이 라이브러리가 설치되어 있다면 이미 사용할 준비는 완료된 것입니다(www.dfanning.com 에서 다운로드할 수 있습니다).

예제 파일 읽기

unidata는 웹을 통해 몇 개의 샘플 netCDF 파일을 제공하고 있습니다. 이 중 rhum.2003.nc 파일을 IDLnetURL 클래스를 이용하여 읽어 옵니다. 다운로드에 약간 시간이 소요될 것입니다.


```

rfile='rhum.2003.nc'
site='www.unidata.ucar.edu'
path='software/netcdf/examples/'
oURL=obj_new('IDLnetURL', URL_HOST=site, $
            URL_PATH=path+rfile)
lfile=oURL->get(filename=rfile)
print, lfile ;예 /Users/yi/rhum.2003.nc
obj_destroy, oURL
    
```

NCDF_BROWSER 실행 방법은 다음과 같이 매우 간단합니다.

```
ncdf_browser, lfile
```

처음 브라우저를 열면, File Overview의 Summary가 디폴트로 보여집니다.



ncdf_browser 실행 예. 파일의 전반적인 설명을 보여 주고 있습니다.

전역 속성 둘러 보기

Global Attributes(전역 속성) 폴더도 열어 보세요. 파일


전반에 공통 적용되는 설명들이 항목별로 보이고 있습니다.



전역속성 Description을 조회한 예

NCDF_BROWSER의 좋은 점은 파일을 살펴보다가 필요한 내용은 바로 IDL 메모리로 읽어올 수 있다는 점입니다. 아래쪽 버튼 중에 Read Global Attribute를 클릭해 보세요.

읽고자 하는 속성을 선택할 수 있으며, 속성 변수 이름은 사용자가 자유롭게 설정할 수 있습니다.



전역 속성 읽어오기

변수 이름(Attribute Name) 항목을 desc로 변경하고 'Read Attribute and Leave'를 실행하면 IDL에서 desc 변수를 볼 수 있습니다.

```
print, desc
```

배열 정의(Dimension) 확인

Dimensions 폴더를 열어 보면, netCDF 파일의 배열을 정의하는 값들을 볼 수 있습니다. 이 예제 파일에는 경도축 144개, 위도축 73개, level축 8개, 시간축 365개의 Dimension 요소들이 보입니다.

변수 둘러 보기

가장 많이 쓰이는 폴더는 역시 Variables일 것입니다. 이 예제 파일에서 Variables 폴더를 펼쳐보면, level, lat, lon, time, rhum 변수가 보입니다. level 변수를 펼쳐 보면 Level 이라는 변수와, 변수 속성을 조회할 수 있습니다.

Level은 8개의 요소를 가지는 1차원 배열로 FLOAT 형이며 값의 범위는 300~1000이라는 것을 바로 알 수 있습니다. 이 값이 의미하는 것은 무엇일까요? units 속성을 보면 millibar라고 보입니다. 지상에 해당하는 1000millibar에서 상층의 300millibar까지 8개의 층으로 이루어진 데이터라고 감을 잡을 수 있겠습니다.

Lat(73개 1차원 배열)과 Lon(144개 1차원 배열), 그리고 Time(365개 1차원 배열) 변수도 살펴 보세요.

time 변수는 좀 신비로운 값이 들어 있습니다만 속성들

을 잘 읽어보면 어렵지 않게 이해할 수 있습니다. 데이터 타입은 Double 형이며 365개 요소를 가지는 1차원 배열로 17549208~17557944의 범위를 가진다는군요. units 항목을 보면 1년 1월 1일 0시부터 진행된 시간 (hour) 단위임을 알 수 있습니다. 1년1월1일0시 이후 17549208시 라는 시간은 어떻게 해석해야 할까요?

```
t=julday(1, 1, 1, 17549208, 0, 0)
print, t, format='(C(CYI04CMOI02CDI02CHI02))'
;; 결과는 2003010100
```

2003년 1월 1일 00시를 의미하는 것을 알 수 있습니다.

rhum 변수는 4차원 배열로 크기는 [144, 73, 8, 365]입니다. 앞의 내용들을 보았으므로 배열의 각 차원이 의미하는 것은 쉽게 알 수 있습니다. 속성에서 볼 수 있듯이 scale_factor=0.01, add_offset=302.660이므로 실제 데이터는 rhum*0.01+302.660로 변환하여 사용해야 합니다.

변수 읽기


읽고자 하는 변수를 마우스로 선택하고 Read Variable 버튼을 클릭해 보세요. .

역시 이 창 안에서 모든 변수를 선택할 수 있습니다. 실제 IDL에서 생성될 변수 이름은 자유롭게 지정할 수 있습니다. Read Variable and Leave 버튼을 이용하면 변수 하나만 읽고 창을 닫지만, Read Variable and Stay 버튼을 이용하면 계속해서 다른 변수를 읽을 수 있습니다. 이 버튼을 이용하면 자동으로 다음 변수를 보여 주기 때문에, 이 버튼만 계속 클릭해도 모든 변수를 다 읽어 올 수 있습니다. 원하는 변수를 다 읽었으면 Quit로 빠져 나오면 되겠지요.



변수 읽기 대화창

예제 netCDF 파일로부터 예제 netCDF 파일의 데이터 표출 데이터를 읽었으므로, 2003년 서울과 카이로의 상대습도 비교 그래프를 다음과 같이 그려볼 수 있습니다.



```
scaled_rhum = rhum * 0.01 + 302.660
iplot, indgen(365)+1, scaled_rhum[51, 21, 0, *], $
  title='Rel. Humidity (Seoul vs Cairo)', $
  xtitle='Days of 2003', ytitle='R.H. (%)'
iplot, indgen(365)+1, scaled_rhum[12, 24, 0, *], $
  COLOR=[0, 255, 0], /OVER
```

속성을 포함한 변수 읽기

Read Variable with Attributes를 이용하면 속성을 포함하여 구조체 변수로 데이터를 읽습니다. time 변수를

읽으면 디폴트로 변수 time_struct(변수 이름은 자유롭게 수정할 수 있습니다)를 생성하는데, 이럴 경우 time_struct.data 가 time 변수의 실제 값이 됩니다. 각 구조체의 data 필드가 변수값이 되는 것입니다.

help, time_struct, /STRUCT

```
** Structure <1bf7b6b8>, 6 tags, length=3000, data length=3000, refs=1:
DATA          DOUBLE  Array[365]
UNITS         STRING   'hours since 1-1-1 00:00:00'
LONG_NAME     STRING   'Time'
ACTUAL_RANGE  DOUBLE   Array[2]
DELTA_T       STRING   '0000-00-01 00:00:00'
AVG_PERIOD    STRING   '0000-00-01 00:00:00'
```

파일 통째로 읽기

Read Entire File 버튼을 클릭하면 IDL 구조체 변수에 파일의 모든 내용을 담습니다. 사용자가 변수 이름을 지정하고 Accept 버튼만 클릭하면 간단하게 처리되는데, 예를 들어 a라고 변수 이름을 지정한 경우, IDL 변수 a는 다음과 같은 구조를 가지게 됩니다.

help, a, /STRUCT

```
** Structure <21cdc08>, 8 tags, length=61394448, data length=61394432, refs=1:
FILENAME      STRING   'rhum.2003.nc'
_GLOBAL_ATTR  STRUCT   -> <Anonymous> Array[1]
_DIMENSIONS   STRUCT   -> <Anonymous> Array[1]
LEVEL         STRUCT   -> <Anonymous> Array[1]
LAT           STRUCT   -> <Anonymous> Array[1]
LON           STRUCT   -> <Anonymous> Array[1]
TIME         STRUCT   -> <Anonymous> Array[1]
RHUM         STRUCT   -> <Anonymous> Array[1]
```

이렇게 되면, a 구조체의 rhum 필드(속성을 포함하므로 역시 구조체)의 data 필드가 4차원 배열 변수 rhum의 값이 되겠지요(한 단계가 더 있는 셈입니다). 이렇게 읽은 경우라면, 다음과 같이 rhum 변수에 scale_factor와 add_offset를 적용할 수 있습니다.

```
scaled_rhum = a.rhum.data*a.rhum.scale_factor+ $
  a.rhum.add_offset
```

좋은 소식

ncdf_browser는 HDF4 포맷의 Scientific Datasets도 다룰 수 있습니다. HDF5는 IDL 자체의 H5_BROWSER() 라는 함수를 이용하면 편리하게 접근할 수 있습니다.

NCDF_DATA 클래스

ncdf_browser는 NCDF_DATA라는 클래스를 기반으로 작성되었습니다. 이 클래스를 직접 사용하여, 프로그래밍 스타일로 데이터를 다룰 수도 있습니다.

```
file='rhum.2003.nc'
oNC=obj_new('NCDF_DATA', file)
ga=oNC->ReadGlobalAttr()
att=oNC->ReadAttribute('TITLE')
dim=oNC->ReadDimension('Lat')
rhum=oNC->ReadVariable('rhum')
srhum=oNC->ReadVariableWithAttr('rhum')
all=oNC->ReadFile()
oNC->Browse ;NCDF_BROWSER 와 같은
obj_destroy, oNC
```

각 함수(메소드)가 의미하는 것은 Browser에서 같은 이름의 버튼들이 하는 일과 같습니다.

